

Digital Noise Filter for GOES-9 Visible Channel

KIGAWA Seiichiro* and Dennis CHESTERS**

Abstract

Anomalous noise is observed on Geostationary Operational Environmental Satellite (GOES)-9 visible imagery. Two detectors of the eight visible detectors of the GOES-9 Imager are found to have 10 counts of RMS noise. The noise is periodic, and may be caused from vibration in the bearings of the momentum wheels that provide 3-axis stabilization for the spacecraft (Chesters and Jentoft-Nielsen, 2002). The visible imagery is useful for qualitative applications, such as analysis of large scale weather features, but quantitative products would be severely degraded by the presence of the noise in imagery (Gray, 2001).

The noise also affects landmark and star observation for image navigation. Therefore, it is desired to find a way to remove the noise both for instrument operation and visible image data users.

This report describes the design of a digital filter that would reduce the anomalous noise on the GOES-9 visible imagery. Visual inspection and quantitative analysis for the imagery that was processed by the digital filter were conducted, and they showed that the digital filter reduced the periodic noise. They also suggested that processed imagery would be acceptable for quantitative products.

1. Introduction

It is reported that anomalous noise appeared at a period of 5-6 samples on visible imagery (Chesters and Jentoft-Nielsen, 2002). This study analyzed the GOES-9 visible imagery taken on 8 Jan. 2002, 1420, which is shown in Figure 1. Figure 2 shows enhanced, zoomed-up space counts on the visible imagery that has 10-bit depth. The noise is almost periodic, and then this feature of the noise shows a way to reduce the noise from the imagery.

The visible imagery is useful for qualitative applications, like analysis of large scale weather features, but quantitative products would be severely degraded by the presence of the noise in the imagery (Gray, 2001). For instance, if visible imagery is

sampled spatially and its pixels have only 8-bit depth, the periodic noise may not be visually noticeable. However, the periodic noise makes a wrong correlation between images; thus it must affect the quantitative products, like a cloud-tracked wind product.



Figure 1. GOES-9 Visible Imagery taken 8 Jan. 2002, 1420

This imagery was used to understand the noise and verify the digital noise filter. The size of the imagery is 20836 (east-west) by 10828 (north-south). The

* Meteorological Satellite Center, Office of Preparation for Meteorological Satellite Operations

** NASA / Goddard Space Flight Center

(Received July 24, 2002 Received October 11, 2002)

imagery is oversampled in the east-west direction.




Figure 2. Contrast-enhanced, Zoomed-up Space Counts

Upper: Contrast-enhanced imagery of the western space region of the GOES-9 visible channel on 8 Jan. 2002, 1420. The imagery has 10-bit depth.

Lower: Space counts of the top (Detector #1) and third (Detector #3) lines of the above imagery.

Table 1 shows two key parameters of the periodic noise on the visible imagery. The standard deviation of the noise suggests the intensity of the additional noise that is caused by external sources. The mean period of the noise is obtained from the pattern matching of the visible imagery. These two parameters determined the design of the noise filter.

2. Filter Design

Figure 3 shows a block diagram for the digital filter that is intended to reduce the noise. The filter

contains two functions: a bandpass filter and an amplitude limitation, which make a cascade connection. The bandpass filter works to extract a periodic signal of 5-6 samples from the imagery. The coefficients and frequency response of the bandpass filter are shown in Figure 4. If a sharp response is desired, more filter elements are required.

The gentle curve of the filter response reduces the number of the filter elements (i.e., required input pixels to calculate an output pixel value).

As shown in Table 1, each detector has an individual period of the noise. This has the consequence that the filter should be designed to match all detectors, or that eight different filters should be prepared for eight detectors.




Figure 3. Block Diagram of Digital Noise Filter

A signal of a period of 5 samples is extracted at the bandpass filter, which has its amplitude limited to prevent generating an excess of a correction signal. The original (upper line) and correction (lower line) signals of the imagery are combined at the final stage of the noise filter.

Detector #	1	2	3	4	5	6	7	8
Standard Deviation [counts]	10.5	3.2	2.8	5.6	5.5	11.4	4.9	5.3
Mean Period [samples]	5.7	5.2	5.5	5.0	5.2	5.0	5.1	5.1

Table 1. Noise Features of GOES-9 Visible Channel

These parameters were obtained from the western space region of the GOES-9 visible imagery on 8 Jan. 2002, 1420.




Figure 4. Bandpass Filter Coefficients and Frequency Response

Upper: Schema of the coefficients of the bandpass filter. An output signal is generated by the convolution of an input signal and filter coefficients. Lower: Frequency response of the bandpass filter. A signal of a period of 5 samples passes this filter.




Figure 5. Amplitude Limitation (Saturation) Filter
This filter limits the amplitude of the correction signal to minimize an inappropriate correction on a cloud scene. An output signal saturates at 10 counts in this case.

Since a cloud scene also contains some signals of a period of 5 samples, the bandpass filter generates a wrong correction signal in some cases. An amplitude

limitation filter works to limit the amplitude of the correction signal. The output of this filter saturates at specified counts. This allows letting some noise pass and minimizes an inappropriate correction on the cloud scene. The filter is defined by

$$\text{Output} = \left[1 - \exp\left(-\frac{\text{Input}}{0.75a}\right) \right] \cdot a \cdot \text{sign}(\text{Input}) \quad (1)$$

where the "sign" function shows the sign of a variable, and a is value of the saturation counts shown in Table 2. Attachment A describes how to generate the filters and the nature of the correction signals.

3. Visual Inspection

This section describes noise filter performance from the point of view of seeing the cloud imagery. Figures 6 and 7 visualize the performance of the noise filter by comparing the original and filter-processed visible imagery.

Morning regions are selected to demonstrate the performance of the filter, because the noise is prominent under low light conditions. The periodic noise is reduced significantly as shown in the space scene of Figure 6. It is also reduced on cloudy scenes, but there seem to be artificial cloud patterns in some areas, like the cumulus region of the lower half of Figure 6. Figure 7 shows the performance under low light conditions. The noise is reduced, while the coastline on the left of the imagery is blurry.

Detector #	1	2	3	4	5	6	7	8
a [counts]	30.4	4.9	1.6	14.6	14.3	33.2	12.2	13.6

Table 2. Saturation Counts

The saturation counts, a, of the amplitude limitation filter are three sigma values of the periodic noise.





Figure 6. Visualization of Filter Performance

Upper: Space scene. Periodic noise is reduced on filter-processed imagery (right).
Lower: Morning cloudy scene. Periodic noise is removed without loss of important cloud features. Artificial cloud patterns appeared a little to the south of the middle of the imagery shown by a white arrow due to natural, periodic cloud patterns. The brightness of the scan line indicated by the white arrow and its adjacent scan lines is presented below the imagery by a thick solid line and two solid lines, respectively. A colored arrow indicates the artificial pattern. These imagery were taken on 8 Jan. 2002, 1420.



Original

Filter-processed

Figure 7. Visualization of Filter Performance under Low Light Conditions

This figure shows filter performance at a dawn scene on 8 Jan. 2002, 1420. Since the imagery is contrast-enhanced, the periodic noise is noticeable.

4. Quantitative Verification

This section shows the results of the quantitative analysis of the filter performance. As shown in Table 1, detector #3 has less periodic noise; thus it can be used as a noise-free cloud scene.

Figure 8 shows how to measure the performance of the filter conceptually. Since a space scene contains only information on the noise, a processed space scene like the upper right of Figure 6 shows how the filter works for the noise. The effect of the filter on the cloud scene is measured using the detector #3 data.

Figure 9 shows a comparison between the original and filter-processed noise. The noise reduces to GOES-8 level, so that the filter works properly for the noise.




Figure 8. Conception of Quantitative Analysis of Filter for Detector #1

This chart shows the measurement of performance of Filter #1 that is used for detector #1. The imagery of detector #3 is used as a cloudy scene.




Figure 9. Filter Performance of Noise-only Case
This chart shows the value of the standard deviation

of the space scene (original) and of the processed space scene (filter-processed). The value of the GOES-8 Imager is also shown as a reference. The noise reduces to GOES-8 level.




Figure 10. Filter Performance of Cloud Scene Case
This chart shows the value of the standard deviation of Error without Noise and Error with Noise shown in Figure 8.

The inappropriate correction on the cloud scene due to the filter is shown in Figure 10. Since the cloud scene contains some signals of a period of 5 samples, the filter makes a wrong correction. This causes the appearance of the artificial cloud pattern or blur, such as the coastline of Figure 7. The standard deviation shown in Figure 10 indicates how the filter affects the cloud scene. It is natural that the standard deviation correlates with the saturation counts shown in Table 2. Further analysis is required to know whether the standard deviation (i.e., the error caused by the filter on the cloud imagery) is acceptable or not.

To understand the effect of the periodic noise on the quantitative cloud-tracked wind product, a pattern matching test was conducted. Figure 11 shows the conceptual schema of the pattern matching test. A visible image of Hurricane Luis taken by GOES-9 in 1995 was used as a background scene (leftmost in Figure 11). The periodic noise that was extracted from the GOES-9 image taken on 8 Jan. 2002 was superimposed on the background scene, and two images were generated (Noisy

Images A and B in Figure 11). These two images had different noise patterns. Pattern matching using a cross-correlation method was conducted between the two images. Furthermore, these two noisy images were processed by the digital filter, and another pattern matching was conducted between filter-processed images (rightmost in Figure 11). Consider each pair of images, with noise-added and noise-suppressed, as observations that have a certain time interval to obtain cloud motion winds. Since each pair of images has the same background, the cloud-tracked wind velocity derived from image-to-image correlation should be zero. Therefore, if it is

not zero, it indicates an error of the pattern matching due to the noise, which will cause an error of the cloud motion winds.

Table 3 shows the results of the pattern matching, which suggests that filter-processed images will be useful for quantitative products.

5. Application to Operational Processing

If the digital noise filter described here will apply to operational processing, the following process will be required to distribute the visible image within an acceptable delay. The required process shown in




Figure 11. Concept of Pattern Matching Test

Noises A and B are extracted from the space region of a recent GOES-9 image: 8 Jan. 2002, 1420.

Image A	Image B	RMS Error [m/s]
Noise-free	Noise-free	0.1
Noisy	Noisy	2.3
Noisy + Filter-processed	Noisy + Filter-processed	0.6

Table 3. Summary of Pattern Matching Analysis

An RMS error is calculated on the assumption that the time interval between Images A and B is 15 minutes.

	Instructions per second	Floating operations per pixel
Addition	2.5×10^6	15
Multiplication	2.5×10^6	15
Referring Look Up Table	0.4×10^6	2
Total	5.4×10^6	32

Table 4. Required Instruction and Floating Operations

This table shows only the instruction and floating operations of the digital noise filter. Additional instruction and floating operations are required to code the filter.

Table 4 is realizable by state-of-the-art commercial computers.

The coefficients of the digital noise filter should be variable because there is no guarantee that noise characteristics are stable. The noise characteristics may change with time, instrument temperature, or wheel speed. The best approach for operational processing is that an appropriate set of the filter coefficients is selected every scan using measurement of the noise at space regions. Figure 12 proposes an approach to select the best filter. In this case, the west space region of full earth-disk observation is used for filter selection. The standard deviation and mean period of the noise at the space regions are measured, and determine the filter coefficients.




Figure 12. Proposed Implementation for Operational Use

Both bandpass and saturation filters are set for optimum processing using two parameters: the mean period and standard deviation measured at the space

regions.

6. Conclusion

The digital noise filter that would reduce the anomalous noise on the GOES-9 visible imagery was studied. This study is summarized as follows:

- GOES-9 visible imagery taken on 8 Jan. 2002, 1420, was used
- The filter was designed to have a bandpass filter and an amplitude limitation filter
- The periodic noise is almost removed by the filter
- The filter makes an artificial cloud pattern or blurs the cloud scene
- Pattern matching analysis suggested that filter-processed imagery would be acceptable for quantitative products.

Acknowledgments. The authors would like to acknowledge Dr. Michael Weinreb of the National Oceanic and Atmospheric Administration / National Environmental Satellite, Data, and Information Service (NOAA/NESDIS) and an anonymous referee for their advice and encouragement.

Reference

Chesters, D., and M. Jentoft-Nielsen, "An examination of coherent noise in GOES-9 visible imagery," 30

Jan. 2002

Gray, D., "GOES-9 Data and Product Quality Assessment," NOAA/NESDIS internal memo, 7 Dec. 2001.

Attachment A. Operational Filter Generation

This attachment describes how to generate operational filters for GOES-9 Imager visible imagery.

A.1. Mean Period and Standard Deviation Measurement

The mean period, τ , and standard deviation, σ , of the visible periodic noise are measured at space regions prior to scanning the earth. They are given by

$$\bar{x} = \frac{1}{N} \sum_{n=0}^{N-1} x_n \quad (A1)$$

$$\sigma = \sqrt{\frac{1}{N} \sum_{n=0}^{N-1} (x_n - \bar{x})^2} \quad (A2)$$

$$b_{ij} = \sum_{m=0}^M (x_{m+i \cdot J} - x_{m+i \cdot J+j})^2, \quad i = 0, 1, \dots, I, \\ j = 1, 2, \dots, J \quad (A3)$$

$$\tau = \frac{1}{I+1} \frac{\sum_{i=0}^I \sum_{j=1}^J \delta[b_{ij} - \min(b_{i1}, b_{i2}, b_{i3}, \dots, b_{ij})] \cdot j}{\sum_{i=0}^I \sum_{j=1}^J \delta[b_{ij} - \min(b_{i1}, b_{i2}, b_{i3}, \dots, b_{ij})]} \quad (A4)$$

where

- x : raw GVAR counts
- I : the number of period measurements
- J : maximum measurable period
- M : the number of pixels for pattern matching
- N : the number of pixels for standard deviation calculation
- $\delta []$: delta function
- min() : minimum value of a dimension

Each detector shows individual noise features, so that eight sets of these parameters should be prepared. The parameters should be stored for monitoring the trend of the noise.

A.2. Bandpass Filter Generation

The bandpass filter is derived from two lowpass filters that have different cutoff frequencies. The cutoff frequencies are given by

$$f_{\text{LOW}} = 2\pi \left(\frac{1}{\tau} - 0.05 \right) \quad (A5)$$

$$f_{\text{HIGH}} = 2\pi \left(\frac{1}{\tau} + 0.05 \right) \quad (A6)$$

Two lowpass filters, LPF_{LOW} and LPF_{HIGH} , are expressed as

$$\text{LPF}_{\text{LOW}}(n) = \begin{cases} \frac{f_{\text{LOW}}}{\pi}, & n = 0 \\ \frac{\sin(f_{\text{LOW}} \cdot n)}{\pi \cdot n} \left[0.54 + 0.46 \cos\left(\pi \frac{n}{15}\right) \right], & n \neq 0 \end{cases} \quad (A7)$$

$$\text{LPF}_{\text{HIGH}}(n) = \begin{cases} \frac{f_{\text{HIGH}}}{\pi}, & n = 0 \\ \frac{\sin(f_{\text{HIGH}} \cdot n)}{\pi \cdot n} \left[0.54 + 0.46 \cos\left(\pi \frac{n}{15}\right) \right], & n \neq 0 \end{cases} \quad (A8)$$

where n is an integer number ranging from -15 to +15. The bandpass filter, BPF, is given by

$$\text{BPF}(n) = \frac{\text{LPF}_{\text{HIGH}}(n)}{\sum_{m=-15}^{15} \text{LPF}_{\text{HIGH}}(m)} - \frac{\text{LPF}_{\text{LOW}}(n)}{\sum_{m=-15}^{15} \text{LPF}_{\text{LOW}}(m)} \quad (A9)$$

The above equation essentially shows that subtracting one lowpass filter from another generates the bandpass filter.




Figure A1. Periodic, contrasty scene like cloudy regions

A.3. Saturation Filter Generation

The saturation counts, a , of the saturation filter should be three times the standard deviation defined in section A.1.

A.4. Nature of Correction Signal

To understand the nature of the correction signal, four sinusoidal waves with different periods and amplitudes are visualized in Figures A1 and A2. Each chart plots GVAR counts in the vertical axis and relative pixel position in the horizontal axis. Charts in sets of three vertically: INPUT, CORRECTION SIGNAL, and OUTPUT show how the filters

process an input signal. The left three charts both in Figures A1 and A2 visualize a process of a sinusoidal wave with a period of five pixels, and the right three a period of eight pixels.

The bandpass filter shown in Figure 4 is used to process the input signals, and the value of the saturation counts, a , is 21.2, which is a value three times the standard deviation of Figure A2 input signals. Since a sinusoidal wave in the region of $1/4 - 1/6.7$ [1/pixel] passes the bandpass filter, roughly speaking, the input signal of the left side cases (i.e., a frequency of $1/5$) passes the bandpass filter. This is




Figure A2. Periodic, less contrasty scene

shown in the output of the bandpass in correction signal charts (circle markers in middle charts). On the right side of the figures, the input is out-of-band on the bandpass filter, and then the bandpass output is almost zero. Therefore, the output of the noise filter is nearly equal to the input.

Let think that the input signals in Figure A1 represent a part of the visible cloudy regions. It can be read from the figure that some cloud features disappear if the saturation filter is not used. This is the reason the saturation filter is employed, and it prevents cloud information loss.

Figure A2 shows filter performance in case the

amplitude of sinusoidal waves is relatively small as compared with the saturation counts. Since the input shown at the upper left is similar to observed periodic noise (e.g., as shown in Figure 2), it can be proved that the noise filter almost clears away the periodic noise as shown at the lower left. Note that the noise filter also clears away cloud features if the frequency and amplitude of the cloud features are nearly equal to those of the periodic noise.

A.5. Software Implementation

The following is a sample source code written in FORTRAN to implement the noise filter operation-

ally. The code shows only an essential portion of the noise filter. This is a case where the space look of the Imager occurs at the west.

```

SUBROUTINE NOISEFILTERWEST(X,Y,LS,CJ,TAULOW,TAUHIGH,
. TAUNOMINAL,SIGMANOMINAL)
C
C      - - - - - SUBROUTINE ARGUMENTS - - - - -
C      INTEGER X(0:14006)
C              input GVAR counts
C      INTEGER Y(0:14006)
C              output GVAR counts
C      INTEGER LS
C              standard deviation high limit
C              (filter processing is bypassed if LS=0) [COUNTS]
C      INTEGER CJ
C              maximum measurable period < 20 [PIXELS]
C      REAL*4 TAULOW
C              mean period low limit [PIXELS]
C      REAL*4 TAUHIGH
C              mean period high limit [PIXELS]
C      REAL*4 TAUNOMINAL
C              nominal value of mean period [PIXELS]
C      REAL*4 SIGMANOMINAL
C              nominal value of standard deviation [COUNTS]
C
C      - - - - - LOCAL VARIABLES - - - - -
C      REAL*4 PI
C              the ratio of the circumference of a circle
C              to its diameter
C      INTEGER N
C              the number of pixels for standard deviation
C              measurement [PIXELS]
C      REAL*4 XMEAN
C              mean GVAR counts [COUNTS]
C      REAL*4 XSUM
C              the sum of GVAR counts [COUNTS]
C      REAL*4 SIGMA
C              standard deviation of GVAR counts [COUNTS]
C      INTEGER SF(0:2046)
C              saturation filter
C      INTEGER CI
C              the number of period measurements < 100 [PIXELS]
C      INTEGER CM
C              the number of pixels for pattern matching [PIXELS]
C      REAL*4 TAU
C              mean period
C      REAL*4 FLOW
C              cutoff frequency (low)
C      REAL*4 FHIGH
C              cutoff frequency (high)
C      REAL*4 LPFLOW(0:30)
C              lowpass filter (low)

```

```

REAL*4 LPFHIGH(0:30)
C                                     lowpass filter (high)
REAL*4 LPFLOWSUM
C                                     the sum of lowpass filter coefficients (low)
REAL*4 LPFHIGHSUM
C                                     the sum of lowpass filter coefficients (high)
REAL*4 BPF(0:30)
C                                     bandpass filter
REAL*4 CS
C                                     correction signal
INTEGER*8 B(0:100,1:20)
C                                     matching results
INTEGER*8 B0(0:100)
C                                     matching results minimum
INTEGER I, J, M, DSUM, DNSUM
REAL*4 TSUM
C                                     local
C
PI = 3.141592653
N = 50
CI = 10
CM = 30
C
C                                     ----- mean GVAR counts -----
XSUM = 0.
DO 1000 I = 0, N-1
    XSUM = XSUM + REAL(X(I))
1000 CONTINUE
XMEAN = XSUM / REAL(N)
C                                     ----- the standard deviation of GVAR counts -----
XSUM = 0.
DO 1100 I = 0, N-1
    XSUM = XSUM + (REAL(X(I))-XMEAN)*(REAL(X(I))-XMEAN)
1100 CONTINUE
SIGMA = SQRT(XSUM/REAL(N))
C                                     ----- saturation filter -----
DO 1200 I = 0, 2046
    SF(I) = FLOOR(SIGN((1-EXP(-ABS(REAL(I-1023)/(3.*SIGMA*0.75))))
    *3.*SIGMA,REAL(I-1023))+0.5)
1200 CONTINUE
C                                     ----- mean period -----
DO 1300 I = 0, CI
    B0(I) = 999999999
    DO 1400 J = 1, CJ
        B(I,J) = 0
        DO 1500 M = 0, CM
            B(I,J) = B(I,J) + (X(M+I*CJ)-X(M+I*CJ+J))*
            (X(M+I*CJ)-X(M+I*CJ+J))
1500 CONTINUE
            IF (B(I,J).LT.B0(I)) THEN
                B0(I) = B(I,J)
            ENDIF
1400 CONTINUE
1300 CONTINUE
TSUM = 0
DO 1600 I = 0, CI
    DSUM = 0
    DNSUM = 0
    DO 1700 J = 1, CJ
        IF (B(I,J).EQ.B0(I)) THEN
            DSUM = DSUM + J
            DNSUM = DNSUM + 1
        ENDIF
1700 CONTINUE
    TSUM = TSUM + REAL(DSUM)/REAL(DNSUM)
1600 CONTINUE
TAU = TSUM/REAL(CI+1)
C
C ===== record tau and sigma into a history file =====
C
C                                     ----- irregular data :
C                                     nominal parameter are used for filters -----
IF (SIGMA.GT.LS .OR. TAU.LT.TAULOW .OR. TAU.GT.TAUHIGH) THEN
    SIGMA = SIGMANOMINAL
    TAU = TAUNOMINAL

```

```

ENDIF
C
C          .          ----- noise filter -----
IF (LS.GT.0.) THEN
C          .          ----- cutoff frequencies -----
          FLOW = 2.*PI*(1./TAU-0.05)
          FHIGH = 2.*PI*(1./TAU+0.05)
C          .          ----- lowpass filters -----
          LPFLOW(15) = FLOW/PI
          LPFHIGH(15) = FHIGH/PI
          DO 1800 I = 1, 15
              LPFLOW(I+15) =SIN(FLOW*REAL(I))*
                  (0.54+0.46*COS(PI*REAL(I)/15.))/(PI*REAL(I))
              LPFLOW(15-I) =SIN(FLOW*REAL(I))*
                  (0.54+0.46*COS(PI*REAL(I)/15.))/(PI*REAL(I))
              LPFHIGH(I+15)=SIN(FHIGH*REAL(I))*
                  (0.54+0.46*COS(PI*REAL(I)/15.))/(PI*REAL(I))
              LPFHIGH(15-I)=SIN(FHIGH*REAL(I))*
                  (0.54+0.46*COS(PI*REAL(I)/15.))/(PI*REAL(I))
1800          CONTINUE
          LPFLOWSUM = 0.
          LPFHIGHSUM = 0.
          DO 1900 I = 0, 30
              LPFLOWSUM = LPFLOWSUM + LPFLOW(I)
              LPFHIGHSUM = LPFHIGHSUM + LPFHIGH(I)
1900          CONTINUE
C          .          ----- bandpass filter -----
          DO 2000 I = 0, 30
              BPF(I) = LPFHIGH(I)/LPFHIGHSUM - LPFLOW(I)/LPFLOWSUM
2000          CONTINUE
C          .          ----- noise filter main process -----
          DO 2100 I = 15, 14006-15
              CS = 0.
              DO 2200 J = 0, 30
                  CS = CS+BPF(J)*REAL(X(I+J-15))
2200          CONTINUE
              Y(I) = X(I)-SF(1023+FLOOR(CS+0.5))
2100          CONTINUE
          DO 2300 I = 0, 14
              Y(I) = X(I)
              Y(14006-I) = X(14006-I)
2300          CONTINUE
C          .          ----- bypassing noise filter -----
ELSE
C          .
          DO 2400 I = 0, 14006
              Y(I) = X(I)
2400          CONTINUE
C          .
ENDIF
C          .
RETURN
END

```

GOES - 9 可視チャンネル用デジタル雑音フィルタ

木川誠一郎*

Dennis Chesters**

GOES - 9 の可視画像に変則的な雑音が観測されている。GOES - 9 イメージャが使用する 8 個の可視検出器のうち、2 個の検出器で強度が 10 カウント (RMS) の雑音が観測されている。その雑音は周期的であり、衛星の姿勢を安定させるモーメントムホイールのベアリングの振動が原因かもしれないと考えられている。可視画像は総観規模の解析のような定性的な利用には有用であるが、雑音の存在によって定量的なプロダクトは大きな精度低下が予想される。

雑音は画像のナビゲーションに使用するランドマークや星位置観測にも影響を与える。そのため、イメージャの運用と可視画像の利用のために雑音を除去する手法の開発が望まれる。

この報告では、GOES - 9 可視画像の変則的な雑音を低減するデジタルフィルタの設計を述べる。デジタルフィルタによって処理された画像の目視と定量的な解析が行われ、デジタルフィルタが周期的な雑音を低減し、処理された画像が定量的なプロダクトにも利用可能であることが示された。

* 気象衛星センター 気象衛星運用準備室

** 米国航空宇宙局 ゴダード宇宙飛行センター